

ASSIGNMENT 3 SOLUTIONS

HPC1 Fall 2013

Due Date: *Thursday, October 17*

(please submit your report electronically to the instructor via email, as one PDF file named *hw3-yourUBitname.pdf*)

Problem 1: Write a program to compute π by the summation:

$$\frac{\pi}{4} = \sum_{i=0}^{N \rightarrow \infty} \frac{(-1)^i}{2i+1},$$

and use **OpenMP** to parallelize the code. Determine the performance of your code as a function of N terms in the sum, and N_p processors. Note that, depending on the granularity of your timer, it may well be necessary to repeat the calculation (say j times, such that jN is a convenient timing interval) and time the total to get reliable average times, especially for smaller values of N . Plot the execution time, parallel speedup, and parallel efficiency as a function of N_p (note the utility of logarithmic scales!). Make careful note of what machine type you are using to perform this study, as for best comparative results versus **MPI** you will want to be consistent in your choice for the second problem.

Solution:

In this problem we are asked to compute the value of π in parallel using partial sums and the summation:

$$\frac{\pi}{4} = \sum_{n=1}^{\infty} \frac{(-1)^{n+1}}{(2n-1)}, \quad (1)$$

which we can easily break up in parallel - let i be the processor index, N the total number of terms in the full sum, then the partial sum for processor i becomes:

$$\frac{\pi^{(i)}}{4} \simeq \sum_{j=i(N/N_p)+1}^{(i+1)(N/N_p)} \frac{(-1)^{j+1}}{(2j-1)}, \quad i \in [0, N_p - 1] \quad (2)$$

where N_p is the number of parallel processes, and we will enforce the rule that N can be divided exactly by N_p , i.e. $\text{mod}(N, N_p) = 0$. Using **OpenMP** you can simply parallelize the summation with a *reduction* clause on the `parallel do` construct, as in the following code.

N	Sequential Time [sec]
128	0.70389E-06
1024	0.50578E-05
1024×10^1	0.50069E-04
1024×10^2	0.49685E-03
1024×10^3	0.49889E-02
1024×10^4	0.49770E-01
1024×10^5	0.49793E+00
1024×10^6	0.49792E+01

Table 1: Optimal sequential times on the 8-core 2.26GHz nodes, Intel compiler 12.1 used with OpenMP turned off, `-O3 -vec_report3 -fpp` options

Table 1 gives the optimal sequential times for various term counts that we will use for measuring parallel speedup (recall that the numerator in the expression for speedup is the optimal sequential time, **not** the parallel time on a single thread or single process). For OpenMP there is little overhead, so the sequential time is only about 0.5% faster on the smaller number of terms. It will be more significant in the MPI case.




Figure 1: OpenMP execution time as a function of N and OMP_NUM_THREADS on a dedicated 8-core L5520 2.26GHz compute node in **UB/CCR**.

Figure 1 shows the execution times as a function of N and OMP_NUM_THREADS on a dedicated 8-core L5520 2.26GHz compute node in the **UB/CCR** cluster, while Figure 3 shows the parallel speedup and efficiency.

Note that in the execution times there is only a relatively substantial overhead for OpenMP when dealing with smaller numbers of terms in the sum. Overall the parallel speedup for the same values of N) is excellent, gaining near 100% efficiency for $N > 102400$. We will have an opportunity to compare the overhead with a message- passing implementation in the next problem.




Figure 2: OpenMP parallel speedup as a function of N and OMP_NUM_THREADS on a dedicated 8-core 2.26GHz compute node in **UB/CCR**.




Figure 3: OpenMP parallel speedup and efficiency as a function of N and OMP_NUM_THREADS on a dedicated 8-core L5520 2.26GHz compute node in **UB/CCR**.

```

1  MODULE myconst
2      integer,parameter :: sp=KIND(1.0),dp=SELECTED_REAL_KIND(2*PRECISION(1.0_sp)), &
3          si=KIND(1),di=SELECTED_INT_KIND(2*RANGE(1_si))
4      real(kind=dp),parameter :: pi=3.141592653589793238462643_dp,pi_over_4 = pi/4.0_dp
5  END MODULE myconst
6
7  PROGRAM PIomp
8      !
9      ! program to use partial sums to compute pi using OpenMP
10     !
11     !$ USE omp_lib
12     USE myconst
13     implicit none
14     !
15     integer (kind=di) :: Nterms    ! number of terms in full sum
16     integer (kind=di) :: Nruns
17     !
18     integer (kind=di) :: i,my_low,my_high,my_sign,Nreps,Nperproc
19     integer :: irep,irun
20     real(kind=dp) :: partial_sum_p,partial_sum_m,sum,time_delta,t0,t1
21     !
22     ! MPI
23     !
24     integer myid,Nprocs
25
26     real(kind=dp) :: dsecnd
27     external dsecnd
28
29     print*, 'range of si,di = ',RANGE(myid),RANGE(Nterms)
30     !
31     ! Do a whole series for Nprocs determined by 10^{i+2}, i=1,Nruns
32     !
33     myid = 0
34     Nprocs = 1
35     Nruns = 8
36     Nterms=128
37     if (myid == 0) then
38         write(*, '(2a12,a9,2a13)') "Nterms","Nperproc","Nreps","error","time/rep"
39     end if
40
41     do irun =1,Nruns
42         !
43         ! make sure that Nterms*Nprocs == 0
44         !
45         ! Nreps - if summing each term takes, say, 10 flop/s, we need
46         !           on the order of 10^9 terms to get a few seconds of
47         !           computation
48         Nreps = 10000000/Nterms
49         if (Nreps == 0) Nreps = Nreps+1
50         !
51         !
52         !
53         time_delta = 0.0
54         do irep = 1,Nreps
55             partial_sum_p = 0.0_dp
56             partial_sum_m = 0.0_dp
57             !
58             !$OMP PARALLEL DEFAULT(SHARED) PRIVATE(i,myid,Nprocs) &
59             !$OMP REDUCTION(+:partial_sum_p,partial_sum_m)
60             !
61             myid = OMP_GET_THREAD_NUM()
62             !$ Nprocs = OMP_GET_NUM_THREADS()
63             Nperproc = Nterms/Nprocs
64             Nterms = Nterms + MOD(Nterms,Nprocs)
65             #ifdef _OPENMP
66                 t0 = OMP_GET_Wtime()
67             #else
68                 t0 = dsecnd()
69             #endif
70             !
71             ! Note that I sum alternating terms separately to avoid
72             ! roundoff errors from adjoining terms and to facilitate
73             ! vectorization of this loop
74             !
75             !$OMP DO
76             do i=1,Nterms,2
77                 partial_sum_p = partial_sum_p + 1.0_dp/(2.0_dp*i-1.0_dp)
78                 partial_sum_m = partial_sum_m - 1.0_dp/(2.0_dp*i+1.0_dp)
79             end do
80             #ifdef _OPENMP
81                 t1 = OMP_GET_Wtime()
82             #else
83                 t1 = dsecnd()
84             #endif
85             !
86             !$OMP END PARALLEL
87             !
88             sum = partial_sum_p + partial_sum_m
89             time_delta = time_delta + (t1-t0)
90         end do
91         time_delta = time_delta/Nreps
92         if (myid == 0) then
93             write(*, '(2i12,i9,2e13.5)') Nterms,Nperproc,Nreps,sum-pi_over_4,time_delta
94         end if
95         if (irun > 1) then
96             Nterms = Nterms*10
97         else
98             Nterms = Nterms*8
99         endif
100     end do
101 END PROGRAM PIomp

```

Problem 2: Repeat problem 1 using **MPI** instead of **OpenMP**.

Hint: simple pseudo-code for splitting the sum into P partial sums might look something like:

```

myID = MyProcNumber()
Np = TotalProcNumber()
mySum = 0
do i=myID*(NsumTerms/Np)+1, (myID+1)*(NsumTerms/Np), 2
    mySum = mySum + 1.0/(2*i-1)
    mySum = mySum - 1.0/(2*i+1)
end do
CollectPartialSums(S)

```

Solution:

We are asked to perform the same computation of π as in the previous problem, but now using message passing (MPI). The final sum is carried out by a single call to **MPI_REDUCE**, otherwise the work can be divided nearly perfectly into the number of processes. So a simple performance model would look like:

$$\tau(N, N_p) = \tau_{red}(8B, N_p) + \tau_{fpdiv} \frac{N}{N_p} \quad (3)$$

where τ_{fpdiv} is the cost of the floating point addition and division in the computation of the partial sums, and $\tau_{red}(8B, N_p)$ is the cost of the reduction, which we can approximate as

$$\tau_{red}(8B, N_p) \simeq \tau_{lat} \log_2(N_p). \quad (4)$$

So in this case we have a speedup,

$$S(N, N_p) = N_p \left[1 + \left(\frac{\tau_{lat}}{N \tau_{fpdiv}} \right) N_p \log_2 N_p \right]^{-1} \quad (5)$$

and the efficiency is then just

$$\mathcal{E}(N, N_p) = \left[1 + \left(\frac{\tau_{lat}}{N \tau_{fpdiv}} \right) N_p \log_2 N_p \right]^{-1} \quad (6)$$

Running the code for N in powers of ten from 10^3 to 10^{10} results in Figure 4.

Note that the MPI execution times are similar to the times we saw for the OpenMP code, but with a little more overhead on the smaller sums. Plotting the speedup will give us a little more insight into any subtleties in the behavior.

From the figure, we can easily discern the limits of our performance model - for small N , the communication costs dominate, and we see little to no speedup. On the opposite side, for large N , the communication costs are relatively minimized, and we observe a near-linear speedup. We could use our performance model to justify this assertion more quantitatively, and this behavior is well illustrated in Figures 4-6.

Note that MPI communication overhead dominates for the lowest range in the number of terms, and grows steadily worse (as you would expect) as the number of processes increases. Compare with the **OpenMP** implementation in the previous problem, which is able to achieve performance gains on smaller amounts of computation, thanks to lower communication overhead.




Figure 4: Execution time for calculation of π using MPI on the 8-core L5520 2.26GHz nodes in **UB/CCR**.




Figure 5: Parallel speedup for calculation of π using MPI on the 8-core L5520 2.26GHz nodes in **UB/CCR**.




Figure 6: Parallel efficiency for calculation of π using MPI on the 8-core L5520 2.26GHz nodes in **UB/CCR**. Note that the efficiency is significantly impacted on smaller sums, even on a single MPI process.

```

1  MODULE myconst
2      integer,parameter :: sp=KIND(1.0),dp=SELECTED_REAL_KIND(2*PRECISION(1.0_sp)), &
3          si=KIND(1),di=SELECTED_INT_KIND(2*RANGE(1_si))
4      real(kind=dp),parameter :: pi=3.141592653589793238462643_dp,pi_over_4 = pi/4.0_dp
5  END MODULE myconst
6
7  PROGRAM PImpi
8      !
9      ! program to use partial sums to compute pi
10     !
11     USE MPI
12     USE myconst
13     implicit none
14     !
15     integer (kind=di) :: Nterms      ! number of terms in full sum
16     integer (kind=di) :: Nruns
17     !
18     integer (kind=di) :: i,my_low,my_high,my_sign,Nreps,Nperproc
19     integer :: irep,irun
20     real(kind=dp) :: sum,partial_sum,partial_sum_p,partial_sum_m,time_delta,t0,t1
21     !
22     ! MPI
23     !
24     character(len=MPI_MAX_PROCESSOR_NAME) :: procname
25     integer myid,Nprocs,length_procname,ierr
26
27     real(kind=dp) :: dsecnd
28     external dsecnd
29
30     CALL MPI_INIT(ierr)
31     if (ierr /= 0) then
32         print*, 'Unable to MPI_Init'
33         STOP
34     end if
35     CALL MPI_COMM_RANK(MPI_COMM_WORLD,myid,ierr)
36     CALL MPI_COMM_SIZE(MPI_COMM_WORLD,Nprocs,ierr)
37     CALL MPI_GET_PROCESSOR_NAME(procname,length_procname,ierr)
38     write(*,'(a24,i4,a4,i4,1x,a32)') 'Greetings from proc ',myid,' of ',Nprocs,procname
39
40     !print*, 'range of si,di = ',RANGE(myid),RANGE(Nterms)
41     !
42     ! Do a whole series for Nprocs determined by 10^{i+2}, i=1,Nruns
43     !
44     Nruns = 9
45     Nterms=128
46     if (myid == 0) then
47         write(*,'(2a12,a9,2a13)') "Nterms","Nperproc","Nreps","error","time/rep"
48     end if
49
50     do irun =1,Nruns
51         !Nterms = Nterms*10
52         !
53         ! make sure that Nterms%Nprocs == 0
54         !
55         !
56         ! Nreps - if summing each term takes, say, 10 flop/s, we need
57         !           on the order of 10^7 terms to get a few seconds of
58         !           computation
59         Nreps = 100000000/Nterms
60         if (Nreps == 0) Nreps = Nreps+1
61         !
62         !
63         !
64         time_delta = 0.0
65         Nperproc = Nterms/Nprocs
66         my_low=myid*Nperproc+1
67         my_high=(myid+1)*Nperproc
68         if (MOD(Nterms,Nprocs).ne.0) then
69             if (myid.eq.0) write(*,*) 'Error, leftovers!'
70             CALL MPI_Abort(MPI_COMM_WORLD,1,ierr)
71         endif
72         do irep = 1,Nreps
73             t0 = MPI_Wtime()
74             ! to take advantage of vectorization, we split the sums
75             ! to avoid vector dependence
76             partial_sum_p = 0.0_dp
77             partial_sum_m = 0.0_dp
78             do i=my_low,my_high,2
79                 partial_sum_p = partial_sum_p + 1.0_dp/(2.0_dp*i-1.0_dp)
80                 partial_sum_m = partial_sum_m - 1.0_dp/(2.0_dp*i+1.0_dp)
81             end do
82             partial_sum = partial_sum_p + partial_sum_m
83             CALL MPI_REDUCE(partial_sum,sum,1,MPI_DOUBLE_PRECISION,MPI_SUM,0, &
84                 MPI_COMM_WORLD,ierr)
85             t1 = MPI_Wtime()
86             time_delta = time_delta + (t1-t0)
87         end do
88         time_delta = time_delta/Nreps
89         if (myid == 0) then
90             write(*,'(2i12,i9,2e13.5)') Nterms,Nperproc,Nreps,sum-pi_over_4,time_delta
91         end if
92         if (irun>1) then
93             Nterms = Nterms*10
94         else
95             Nterms = Nterms*8
96         endif
97     end do
98     CALL MPI_FINALIZE(ierr)
99 END PROGRAM PImpi

```